

(DRAFT)

Notes on Formulating and Solving Computable General Equilibrium Models within GAMS

by

Dhazn Gillig
Research Scientist
Department of Agricultural Economics
College Station, TX
dgillig@taexgw.tamu.edu
(979) 845-3153

Bruce A. McCarl
Professor
Department of Agricultural Economics
College Station, TX
mccarl@tamu.edu
(979) 845-1706

July 2002

Notes on Solving Computable General Equilibrium Models in GAMS

GAMS is commonly used to solve computable general equilibrium (CGE) models. This document presents background material on CGE modeling with GAMS first covering the general structure of a CGE and then focusing on the GAMS formulation and subsequent solution of the problem. Note in this coverage we will not be using the MPSGE approach as we prefer the more structural approach. Readers wishing details on MPSGE should consult the materials developed by Rutherford on <http://debreu.colorado.edu/mainpage/mpsge.htm>

The majority of our discussion will be presented in the context of a two goods, two household, two factor setting in a closed economy. This setting is adapted from an example provided in GAMS model library (TWO3MCP) which is based on the work of Shoven and Whalley (1984).

CGE Model Background

CGE models depict entire economies. In such models prices and production/consumption in both factor and output markets are endogenous as is income. At equilibrium there are several characteristics of CGE model solutions:

1. The total market demand equals the total market supply for each and every factor and output market.
2. Prices are set so that equilibrium profits of firms are zero with all rents accruing to factors.
3. Household incomes equal household expenditures.
4. Government tax revenues equal government expenditures including subsidy payments.

Basic Notation

From here on we assume a perfectly competitive economy that produces a number of goods denoted by j which contains a number of households denoted by h using the basic factor inputs labor and capital. The production quantities, factor usages, and associated prices are defined as follows:

- Q_j is the production of goods by the j^{th} sector,
- P_j is the price of goods produced by the j^{th} sector,
- L_j is the usage of labor by the j^{th} sector,
- W_L is the price of labor (the wage rate),
- $a_{j1,j}$ is the amount of goods used from sector $j1$ when producing one unit of goods in the j^{th} sector,
- K_j is the usage of capital in the j^{th} sector,
- W_K is the price of capital, and
- X_{jh} is the consumption of the goods produced by sector j by household h .

We also assume that the factors are all owned by households where

- \bar{L}_h is the endowment of labor in household h , and
- \bar{K}_h is the endowment of capital in household h .

Equilibrium Conditions (without Taxes)

A set of non-zero prices P_j , W_L , and W_K , consumption levels X_{jh} , production levels Q_j and factor usages L_j , and K_j constitutes an economic equilibrium solution (also called a Walrasian equilibrium) and a solution to a CGE of the situation if

- I. The total demand is less than or equal to the total supply in every factor market.

The total supply is the sum across the household endowments; in other words, the excess demand in the factor input markets is less than or equal to zero:

$$\sum_j L_j - \sum_h \bar{L}_h \leq 0 \quad (1)$$

$$\sum_j K_j - \sum_h \bar{K}_h \leq 0. \quad (2)$$

- II. The total demand in every output market including consumer and intermediate production usage is less than or equal to the total supply in that market. In other words, the excess demand in each output market is less than or equal to zero:

$$\sum_h X_{jh} + \sum_{j1} a_{j,j1} Q_{j1} - Q_j \leq 0 \quad \forall j. \quad (3)$$

- III. Profits are zero in each sector:

$$P_j Q_j = \sum_{j1} P_{j1} a_{j1,j} Q_{j1} + W_L L_j + W_K K_j \quad \forall j. \quad (4)$$

Walras' Law also introduces a requirement. Namely,

“Walras' Law: For any price vector \mathbf{P} , we have $\mathbf{PZ}(\mathbf{P}) = 0$; i.e., the value of the excess demand is identically zero.” (Varian, page 317)

where $\mathbf{Z}(\mathbf{p})$ is the excess demand in the factor and product markets as shown in equations (1), (2), and (3). Then we get the additional requirement that

$$\sum_j P_j \left(Q_j - \sum_{j1} a_{j,j1} Q_{j1} - \sum_h X_{jh} \right) + W_L \left(\sum_j L_j - \sum_h \bar{L}_h \right) + W_K \left(\sum_j K_j - \sum_h \bar{K}_h \right) = 0. \quad (5)$$

Walras' Law and equation (5) coupled with an assumption of nonnegative prices and excess demands also imply a complementarity between prices and excess demand (Ferris and Munson, Ferris and Pang, and Manne). Namely, if the total demand is less than the total supply for the commodity j market then the price in that market must be zero. Conversely, prices will be nonzero only if supply equals demand (Varian). For example, for the factors the factor prices must be zero and the factors not all used up or for non zero prices to exist the factors must be all consumed:

$$0 \leq W_L \perp \sum_j L_j - \sum_h \bar{L}_h \leq 0 \quad (6)$$

$$0 \leq W_K \perp \sum_j K_j - \sum_h \bar{K}_h \leq 0 \quad (7)$$

Similarly the product prices must be zero and the products not all consumed or for non-zero prices to exist the products must be all consumed:

$$0 \leq P_j \perp \sum_h X_{jh} + \sum_{j1} a_{j,j1} Q_{j1} - Q_j \leq 0 \quad (8)$$

For the firm, profits must equal zero and a non-zero production level achieved or profits can be less than costs without the firm producing.

$$0 \leq Q_j \perp P_j Q_j \leq \sum_{j1} P_{j1} a_{j1,j} Q_{j1} + W_L L_j + W_K K_j \quad (9)$$

Finally household income is defined by

$$0 \leq Income_h \perp Income_h \geq W_L \bar{L}_h + W_K \bar{K}_h \quad (10)$$

where the \perp notation means that at least one of the adjacent inequalities must be satisfied as an equality. These complementary relationships are important elements of the solution method employed. Namely, we use the solver PATH to impose the complementarity (Ferris and Munson, Ferris and Pang).

Determining Consumer Demand

Household goods consumption is determined by assumptions about consumer behavior. Consumers are generally assumed to maximize utility subject to a budget constraint. The assumed form of the utility is a constant elasticity of substitution (CES) function. Note other functional forms such as Cobb Douglas, Linear Expenditure System, or Translog can also be used. Under the assumed functional form we may now express household demand for goods by deriving household level product demand relations. The CES utility function $[U_h(\cdot)]$ is assumed strictly quasi-concave and differentiable and is given by:

$$U_h = \left[\sum_j (\alpha_{jh})^{1/\sigma_h} (X_{jh})^{(\sigma_h-1)/\sigma_h} \right]^{\sigma_h/(\sigma_h-1)}$$

where

α_{jh} is the expenditure share for consumption spending on goods j for household h , and σ_h is the elasticity of substitution in the household h function.

Assume each consumer operates subject to a budget constraint and has an initial endowment of factors, but no initial endowments of goods. The budget constraint is then

$$\sum_j P_j X_{jh} \leq W_L \bar{L}_h + W_K \bar{K}_h = Income_h$$

where $Income_h$ is household h income. In turn, consumer demand functions can be derived.

Such functions will be homogenous of degree zero in prices implying that a uniform change in all prices will not affect quantities demanded and in turn the equilibrium. This property also requires prices to be normalized as will be discussed later so we have a unique solution. The resultant demand function for commodity j is:

$$X_{jh} = \frac{\alpha_{jh} (Income_h)}{P_j^{\sigma_h} \sum_j (\alpha_{jh} (P_j)^{1-\sigma_h})}$$

In turn, we add an equation to the CGE model of the form

$$0 \leq X_{jh} \perp X_{jh} - \frac{\alpha_{jh} (Income_h)}{P_j^{\sigma_h} \sum_j (\alpha_{jh} (P_j)^{1-\sigma_h})} \leq 0 \quad \forall \quad j \& h \quad (11)$$

Producer Factor Demand

The quantity of factors used in production is determined by assumptions about producer behavior. CGE models are typically built under the assumption that producers are maximizing profits with production characterized by a known production function under constant returns to

scale and a constant elasticity of substitution (CES) function. Also, note that other functional forms such as fixed coefficients or Cobb Douglas can also be used. The CES production function is given by:

$$Q_j = \phi_j \left[\delta_j L_j^{(\sigma_j-1)/\sigma_j} + (1-\delta_j) K_j^{(\sigma_j-1)/\sigma_j} \right]^{\sigma_j/(\sigma_j-1)}$$

where

ϕ_j is the scale parameter for sector j ,

δ_j is another CES function parameter for sector j ,

σ_j is the elasticity of substitution of labor and capital in sector j

Applying the assumption of cost minimization and given that sector j produces quantity Q_j then an expression for factor demand can be derived which is:

$$L_j = \frac{1}{\phi_j} Q_j \left[\delta_j + (1-\delta_j) \left(\frac{\delta_j W_K}{(1-\delta_j) W_L} \right)^{(1-\sigma_j)} \right]^{\sigma_j/(1-\sigma_j)}$$

$$K_j = \frac{1}{\phi_j} Q_j \left[\delta_j \left(\frac{(1-\delta_j) W_L}{\delta_j W_K} \right)^{(1-\sigma_j)} + (1-\delta_j) \right]^{\sigma_j/(1-\sigma_j)}$$

where W_L and W_K are the labor (L) and capital (K) factor prices.

These are included in the CGE model as follows

$$0 \leq L_j \perp \frac{1}{\phi_j} Q_j \left[\delta_j + (1-\delta_j) \left(\frac{\delta_j W_K}{(1-\delta_j) W_L} \right)^{(1-\sigma_j)} \right]^{\sigma_j/(1-\sigma_j)} - L_j \leq 0 \text{ for all } j \quad (12)$$

$$0 \leq K_j \perp \frac{1}{\phi_j} Q_j \left[\delta_j \left(\frac{(1-\delta_j) W_L}{\delta_j W_K} \right)^{(1-\sigma_j)} + (1-\delta_j) \right]^{\sigma_j/(1-\sigma_j)} - K_j \leq 0 \text{ for all } j \quad (13)$$

Incorporation of taxes

Taxes are a fact of life and a CGE should contain tax provisions. Tax treatments range through a number of possibilities (Shoven and Whalley and Creedy). To include taxes, a

modification is needed in the zero profit condition to include tax payments as a cost of doing business. We also need to model a government disposition of tax revenues. We introduce taxes into the CGE assuming that there are taxes on production net income and taxes on household income. Suppose that the tax rates are

$$\begin{aligned} T_j & \text{ percent imposed on corporate net income, and} \\ T_h & \text{ percent imposed on household income.} \end{aligned}$$

Also suppose that each household type can deduct F_h . In turn, total tax revenue (R) collected by the government is

$$R = \sum_j T_j (P_j Q_j - W_L L_j - W_K K_j) + \sum_h T_h (W_L \bar{L}_h + W_K \bar{K}_h - F_h),$$

which applies the corporate tax rate to an accounting of corporate net income and the household income tax to gross revenue less deductions. In turn, we assume the total tax revenue is redistributed to households in the form of transfer payments (TR_h) or is expended on government purchases of goods (GP_j). Assume that government purchases are proportional to the gross revenue at the rate (s_j) as are household transfer payment shares (s_h). Therefore, household h receives $TR_h = s_h * R$, and the government expenditures on the j th sector are $GP_j = s_j * R$ where

$$\sum_h s_h + \sum_j s_j = 1, \text{ and in turn } \sum_h TR_h + \sum_j GP_j = R.$$

This introduces modifications to the model above as follows. First, the market balance expands to include government purchases transformed to be in the quantity unit

$$\sum_h X_{jh} + \sum_{j1} a_{j,j1} Q_{j1} - Q_j + s_j R / P_j \leq 0 \quad (3a)$$

Second, the zero profit condition expands to include tax effects

$$0 \leq Q_j \perp P_j Q_j \leq \sum_{j1} \mathbf{P}_{j1} \mathbf{a}_{j1,j} Q_j + W_L L_j + W_K K_j + T_j (P_j Q_j - \sum_{j1} \mathbf{P}_{j1} \mathbf{a}_{j1,j} Q_j - W_L L_j - W_K K_j) \quad (9a)$$

Third, the household income equation is adjusted to reflect tax incidence and tax revenue redistribution

$$0 \leq Income_h \perp (1 - T_h) * (W_L \bar{L}_h + W_K \bar{K}_h - F_h) + s_h \mathbf{R} \leq Income_h \quad (10a)$$

Fourth, we add a government income balance

$$0 \leq R \perp R \leq \sum_j T_j (P_j Q_j - \sum_{j1} \mathbf{P}_{j1} \mathbf{a}_{j1,j} Q_j - W_L L_j - W_K K_j) + \sum_h T_h (W_L \bar{L}_h + W_K \bar{K}_h - F_h) \quad (14)$$

The complementary condition in equation (14) indicates that the government satisfies its budget constraint when total revenue is positive.

Casting the model into GAMS

Now suppose we turn our attention to GAMS. Here we will set up a model using the basic parameters of a model by Shoven and Whalley (1984) as implemented in the TWO3MCP model in the GAMS model library. That structure depicts an economy with two factors of production (labor and capital), two commodities are produced (food and nonfood) by sectors with those names; and there are two household classes (farm and nonfarm). We assume familiarity with basic GAMS instructions so will introduce the model in GAMS format.

First, we need to define the basic subscripts or sets. We define three sets. The set **Factors** defines the factors of production in the model encompassing labor and capital. The set **Sector** depicts the j subscript representing the production sectors in the model. The set **Households** denotes the set of different household consumption units in the h subscript above.

Then we define the variables. In this case there are 7 as listed below

Symbols Used for Problem Variables

Name in Theoretical Model	Corresponding GAMS Item	Brief Description
W_L, W_K	FACTORPRICE(Factor)	Prices for the factors labor and capital
L_j, K_j	FACTORQUAN(Factor, Sector)	Factor use by a producing sector
P_j	COMPRICE(Sector)	Prices of commodities
X_{jh}	DemCommod(Households, Sector)	Commodity demand by households
Q_j	PRODUCTION(Sector)	Production quantity
$Income_h$	HHIncome(households)	Household income
R	TAXREVENUE	Total government tax revenue

We also need GAMS equations associated with each of the equations developed above and these are:

Symbols Used for Problem Equations

Eqn Number in Model	Corresponding GAMS Item	Brief Description
1,2	FactorMkt(Factor)	Factor market balances
12,13	FactorDem(Factor, Sector)	Commodity demand by households
3a	CommodMkt(Sector)	Commodity market balance
11	CommodDem(Households, Sector)	Commodity demand by households
9a	Profit(Sector)	Zero profit condition
10a	Income(households)	Household budget constraint
14	GovBal	Government budget constraint

To specify this model we also need to define a number of data items. These include

Symbols Used for Problem Data

Symbol Name Model	Corresponding GAMS Item	Brief Description
σ_h	SigmaC(Households)	Elasticity of substitution in household CES
α_{jh}	Alpha(Sector, Households)	Consumption share in household CES

\bar{L}_h, \bar{K}_h	Endowment(Factor,Households)	Household endowments of factors
$a_{jl,j}$	Intermediateuse(Sector1,Sector)	Use of goods in sector1 when producing in sector
ϕ_j	Phi(Sector)	Scale parameter in CES production function
δ_j	Delta(Factor,Sector)	Distribution parameter in CES production
σ_j	Sigma(Sector)	Elasticity of production factor substitution
s_h	TaxShare(Households)	Household share of tax disbursements
s_j	Governmentpurch(Sector)	Government goods purchase dependence on Revenue
T_h	Incometax	Household tax level
T_j	CorporateTax(Sector)	Tax on corporate earnings
F_h	TaxExemption(Households)	Household tax exemptions

The class of models that this CGE falls into is the so called mixed complementarity class (an MCP) in GAMS. To solve such models we must have a model with

- 1) complementarity requirements associated with each equation since each has the requirement denoted by \perp ,
- 2) a consistent set of variable and equation definitions where each variable is of the same dimension as an associated equation that it is complementary with,
- 3) a model where no variable is complementary with more than one equation or vice versa, and
- 4) a model where every variable and equation has a complementary partner.

Our model satisfies these requirements and the complementary relationships are expressed in the model statement. Specifically, in GAMS we express this in the MODEL statement employing the notation involving the periods in the statement below

```
MODEL CGEModel
    /FactorMkt.FACTORPRICE, FactorDem.FACTORQUAN,
    CommodMkt.COMPRICE,commoddem.DemCommod
```

Profit.PRODUCTION, Income.Hhincome, GovBal.TAXREVENUE/;

This notation shows the complementary partners and indicates that the complementary relations in the table below need to be satisfied in the solution.

Complementary Items

Variable Name	Equation Name
FACTORPRICE(Factor)	FactorMkt(Factor)
FACTORQUAN(Factor,Sector)	FactorDem(Factor,Sector)
COMPRICE(Sector)	CommodMkt(Sector)
DemCommod(Households,Sector)	CommodDem(Households,Sector)
PRODUCTION(Sector)	Profit(Sector)
HHIncome(households)	Income(households)
GovtIncome	GovBal

Note how the dimensions of the items are the same.

The variables mentioned above are declared in GAMS using the statement below

POSITIVE VARIABLES

FACTORPRICE(Factor)	Prices for factors
FACTORQUAN(Factor,Sector)	Factor use by a sector
COMPRICE(Sector)	Prices of commodities
DemCommod(Households,Sector)	Demand by households
PRODUCTION(Sector)	Production quantity
HHIncome(households)	Household income
GovtIncome	Government tax revenue

The equations are defined as follows

EQUATIONS

FactorMkt(Factor)	Factor market balances
FactorDem(Factor,Sector)	Factor demand by a sector
CommodMkt(Sector)	Commodity market balance
CommodDem(Households,Sector)	Commodity demand by households
Profit(Sector)	Zero profit condition
Income(households)	Household budget constraint
GovBal	Government budget constraint ;

and are specified as follows

Factor Market Balance

For factors, the factor used in production is less than or equal to the total supply in every factor input market which is composed from the household endowments.

$$\begin{aligned} & \text{FactorMkt(Factor)..} \\ & \text{sum(households,Endowment(Factor,households))} = \text{G=} \\ & \text{sum(Sector,FactorQuan(Factor,Sector))} \quad ; \end{aligned}$$

Commodity Market Balance

For commodities, the total demand in the output market for each sector including consumer demand, intermediate production usage, and government purchases are less than or equal to the total production in that market.

$$\begin{aligned} & \text{CommodMkt(Sector)..} \\ & \text{Production(Sector)} = \text{G=} \\ & \text{sum(households,DemCommod(Households,Sector))} \\ & + \text{Governmentpurch(sector)*GovtIncome/COMPRICE(Sector)} \\ & + \text{sum(othersector,} \\ & \quad \text{intermediateuse(sector,othersector)*production(othersector));} \end{aligned}$$

Firm Zero Profit Condition

For each production sector, revenues are less than or equal to costs with in effect all rents allocated to factors. Thus the total revenue is less than factor usage costs plus costs of intermediate products obtained plus tax payments.

$$\begin{aligned} & \text{Profit(Sector)..} \\ & \text{sum(Factor,FactorPrice(Factor)*FactorQuan(Factor,Sector))} \\ & + \text{sum(othersector,} \\ & \quad \text{ComPrice(otherSector)*intermediateuse(othersector,sector)*production(sector))} \\ & + \text{corporatetax(sector)*} \\ & \quad \text{ComPrice(Sector)* Production(Sector)} \\ & - \text{sum(othersector,} \end{aligned}$$

$$\begin{aligned} & \text{ComPrice(OtherSector)*intermediateuse(Othersector,sector)*production(sector))} \\ & -\text{sum(Factor,FactorPrice(Factor)*FactorQuan(Factor,Sector))} \\ =G= & \text{ComPrice(Sector)* Production(Sector) ;} \end{aligned}$$

Household Income

For each household, the income is less than or equal to the household income from factor sales adjusted for taxes and tax deductions plus transfer payments from the government.

$$\begin{aligned} \text{HHIncome(households) } =G= & \\ & (1-\text{incometax}) * \text{sum(Factor,Endowment(Factor,households) * FactorPrice(Factor))} \\ & + \text{incometax * TaxDeduction(Households)} \\ & + \text{TaxShare(households) *TaxRevenue ;} \end{aligned}$$

Government Tax Income

For the government tax, the government income is less than or equal to taxes from households applied to factor income less deductions plus corporate taxes applied to net income.

$$\begin{aligned} \text{GovBal..} & \\ \text{GovtIncome } =G= & \\ & \text{sum(sector,corporatetax(sector)*} \\ & \text{(ComPrice(Sector)* Production(Sector)} \\ & \text{-sum(OtherSector,} \\ & \text{ComPrice(OtherSector)*intermediateuse(Othersector,sector)} \\ & \text{*production(secto1))} \\ & \text{-sum(Factor,FactorPrice(Factor)*FactorQuan(Factor,Sector))})} \\ & +\text{sum(households,incometax} \\ & \text{* (sum(Factor,Endowment(Factor,households) * FactorPrice(Factor))} \\ & \text{-TaxDeduction(Households))}; \end{aligned}$$

Household Commodity Demand

For each household we assume a CES utility function and derive commodity demand subject to a budget constraint. In turn, this lets us develop a formula for commodity demand that we imbed in the model. Demand is a function of income and prices. The function is also homogeneous of degree zero in prices which will require other actions later.

$$\begin{aligned} & \text{CommodDem}(\text{Households}, \text{Sector}).. \\ & \text{DemCommod}(\text{Households}, \text{Sector}) = E = \\ & \quad (\text{HHIncome}(\text{households}) / \sum(\text{Sector1}, \alpha(\text{Sector1}, \text{households}) \\ & \quad \quad * \text{ComPrice}(\text{Sector1}) ** (1 - \text{SigmaC}(\text{households}))) \\ & \quad) * \text{Alpha}(\text{Sector}, \text{households}) * (1 \\ & \quad / \text{ComPrice}(\text{Sector}) ** \text{sigmaC}(\text{households})); \end{aligned}$$

Producer Factor Demand

For each producer we assume a CES production function and derive factor demand assuming cost minimizing production of a given quantity of product. In turn, this lets us develop a formula for factor demand that we imbed in the model. Demand is a function of production quantity and factor prices.

$$\begin{aligned} & \text{FactorDem}(\text{Factor}, \text{Sector}).. \\ & \text{FACTORQUAN}(\text{Factor}, \text{Sector}) = g = \\ & \quad \text{Production}(\text{Sector}) * \text{Phi}(\text{Sector}) ** (\text{sigma}(\text{Sector}) - 1) \\ & \quad * (\text{Delta}(\text{Factor}, \text{Sector}) \\ & \quad \quad * (\sum(\text{Factor1}, \text{Delta}(\text{Factor1}, \text{Sector}) ** \text{sigma}(\text{Sector}) \\ & \quad \quad \quad * (\text{FactorPrice}(\text{Factor1})) ** (1 - \text{sigma}(\text{Sector}))) \\ & \quad \quad ** (1 / (1 - \text{sigma}(\text{Sector}))) / \text{Phi}(\text{Sector})) / \\ & \quad \quad \text{FactorPrice}(\text{Factor}) ** \text{sigma}(\text{Sector}) \quad ; \end{aligned}$$

Complementarity

Walras's Law introduces a requirement. Namely, for any price vector P, we have $PZ(P) = 0$; i.e., the value of the excess demand, $Z(P)$, is identically zero." (Varian, page 317). This implies price or excess demand = 0; This makes our model complementary in terms of following items

Variable Name	Complementary Equation
FACTORPRICE(Factor)	FactorMkt(Factor)
FACTORQUAN(Factor, Sector)	FactorDem(Factor, Sector)
COMPRICE(Sector)	CommodMkt(Sector)
DemCommod(Households, Sector)	CommodDem(Households, Sector)
PRODUCTION(Sector)	Profit(Sector)
HHIncome(Households)	Income(Households)
TAXREVENUE	GovBal

We reflect that requirement in GAMS as follows

```
MODEL CGEModel /FactorMkt.FACTORPRICE,
                FactorDem.FACTORQUAN,
                CommodMkt.COMPRICE,
                commoddem.DemCommod
                Profit.PRODUCTION,
                Income.Hhincome,
                GovBal.GovtIncome/;
```

Normalizing Prices

The formulation needs to overcome the homogeneous of degree zero property since an infinite number of prices can yield a given solution. This is done either by setting the income for one household equal to one or the price of a commodity to one. We choose the former

```
HHincome.L(households) =
    sum(Factor, FactorPrice.l(Factor) * Endowment(Factor,Households));
HHincome.fx(Households)$ (ord(Households) = 1) = HHincome.l(Households);
```

Starting Points and Bounds

We also use starting points and lower bounds to avoid numerical problems.

Solving

This problem in turn is best solved with the PATH solver. Details on that solver can be found at <http://www.gams.com/solvers/solvers.htm#PATH>. So we choose PATH as the solver and solve with

```
OPTION MCP = PATH;
SOLVE CGEModel USING MCP;
```

The Solution

In turn, a solution arises and you can use normal report writing, graphics etc. as with any other model

```
---- VAR FACTORPRICE  Prices for the factors
```


	LOWER	LEVEL	UPPER	MARGINAL
Labor	0.010	0.931	+INF	.
Capital	0.010	1.162	+INF	.

---- VAR FACTORQUAN Factor use

	LOWER	LEVEL	UPPER	MARG
Labor .Food	.	36.848	+INF	.
Labor .NonFood	.	28.152	+INF	.
Capital.Food	.	10.506	+INF	.
Capital.NonFood	.	16.494	+INF	.

---- VAR COMPRICE Prices of commodities

	LOWER	LEVEL	UPPER	MARGINAL
Food	0.010	0.867	+INF	.
NonFood	0.010	0.977	+INF	.

---- VAR DemCommod Commodity Demand

	LOWER	LEVEL	UPPER	MARGINAL
Farm .Food	.	3.230	+INF	.
Farm .NonFood	.	4.299	+INF	.
NonFarm.Food	.	50.429	+INF	.
NonFarm.NonFood	.	42.155	+INF	.

---- VAR PRODUCTION Production Quantity

	LOWER	LEVEL	UPPER	MARGINAL
Food	.	53.659	+INF	.
NonFood	.	46.454	+INF	.

---- VAR HHIncome Household income

	LOWER	LEVEL	UPPER	MARGINAL
Farm	7.000	7.000	7.000	9.429E-12
NonFarm	0.010	84.895	+INF	.

	LOWER	LEVEL	UPPER	MARGINAL
---- VAR TAXREVENUE	.		0.092	+INF

Calibration

In previous sections we show how to develop a CGE model and implement it in GAMS to solve for the equilibrium solutions of price (P_j , W_L , and W_K), consumption levels (X_{jh}), production levels (Q_j) and factor usages (L_j and K_j) given specific functional forms along with their parameter values. However, this forms a fairly difficult nonlinear system of equations and one must first insure that a solution exists and that the data within the model are consistent. This has come to be known as the calibration or a so-called “benchmark equilibrium” problem in CGE modeling. (Shoven and Whalley, 1992, or Robinson, and El-Said, 2000).

The first step in the calibration problem involves making certain that the benchmark equilibrium data is consistent with the equilibrium conditions for a base year of data. In practice, the benchmark equilibrium data directly obtained from national accounts (input-output tables) or other government agencies are unlikely to be consistent with the general equilibrium conditions (i.e. the producer factor cost spent on labor or capital may not equal the household labor or capital income). As a result, adjustments are needed to ensure that the equilibrium conditions hold. Numerous adjustments can be applied on a case by case basis. Section v in Shoven and Whalley (1992) discusses procedures for constructing and adjusting the benchmark equilibrium data set.

After constructing the benchmark equilibrium data, the next step is to determine parameter values conforming to this benchmark equilibrium data. Because the benchmark equilibrium data mainly provide information on price and quantity, in some cases such additional extraneous data are required. For example, recall the CES production function previously discussed,

$$Q_j = \phi_j \left[\delta_j L_j^{(\sigma_j-1)/\sigma_j} + (1-\delta_j) K_j^{(\sigma_j-1)/\sigma_j} \right]^{\sigma_j/(\sigma_j-1)},$$

this CES production function has 3 parameters, namely ϕ_j , δ_j , and σ_j ; however, the benchmark equilibrium data typically provide only information on the ratios of factor price and factor demand, but do not provide information on the elasticity of substitution. Hence, the value of elasticity of substitution must be exogenously specified. The value of elasticity of substitution can be obtained either from the econometric estimation or from the literature. After the elasticity of substitution is available, the parameters (ϕ_j , and δ_j) associated with the CES production function can be calculated as shown below.

From the factor demands derived previously,

$$L_j = \frac{1}{\phi_j} Q_j \left[\delta_j + (1 - \delta_j) \left(\frac{\delta_j W_K}{(1 - \delta_j) W_L} \right)^{(1 - \sigma_j)} \right]^{\sigma_j / (1 - \sigma_j)},$$

and

$$K_j = \frac{1}{\phi_j} Q_j \left[\delta_j \left(\frac{(1 - \delta_j) W_L}{\delta_j W_K} \right)^{(1 - \sigma_j)} + (1 - \delta_j) \right]^{\sigma_j / (1 - \sigma_j)},$$

we can arrange these into

$$W_L L_j = W_L \frac{1}{\phi_j} Q_j \left[\delta_j + (1 - \delta_j) \left(\frac{\delta_j W_K}{(1 - \delta_j) W_L} \right)^{(1 - \sigma_j)} \right]^{\sigma_j / (1 - \sigma_j)} \quad (15)$$

$$W_K K_j = W_K \frac{1}{\phi_j} Q_j \left[\delta_j \left(\frac{(1 - \delta_j) W_L}{\delta_j W_K} \right)^{(1 - \sigma_j)} + (1 - \delta_j) \right]^{\sigma_j / (1 - \sigma_j)}. \quad (16)$$

Then factor shares for sector j , ω_{KL}^j , and ω_{LK}^j is equal to

$$(16) \div (15), \quad \omega_{LK}^j = \frac{L_j W_L}{K_j W_K} = \frac{\delta_j^\sigma W_K^{1 - \sigma}}{(1 - \delta_j)^\sigma W_L^{1 - \sigma}}, \quad (17a)$$

and
$$\omega_{KL}^j = \frac{K_j W_K}{L_j W_L} = \frac{(1-\delta_j)^\sigma W_L^{1-\sigma}}{\delta_j^\sigma W_K^{1-\sigma}} \quad (17b)$$

With the assumption that the benchmark equilibrium factor prices W_L and $W_K = 1$, then we can solve for δ_j from equation (17a) expressed in term of ω_{LK}^j ,

$$\delta_j = \frac{(\omega_{LK}^j)^{1/\sigma}}{1 + (\omega_{LK}^j)^{1/\sigma}},$$

where $\omega_{LK}^j = L_j / K_j$ and is directly provided from the benchmark equilibrium. Note that by solving equation (17b), δ_j can also be expressed in term of ω_{KL}^j ,

$$\delta_j = \frac{1}{1 + (\omega_{KL}^j)^{1/\sigma}}$$

where $\omega_{KL}^j = K_j / L_j$ and is also directly provided from the benchmark equilibrium. Next, the value of the sector j scale parameter, ϕ_j , can be determined through the zero profit conditions:

$$P_j Q_j = W_L \frac{1}{\phi_j} Q_j \left[\delta_j \left(\frac{(1-\delta_j)}{\delta_j} \right)^{(1-\sigma_j)} + (1-\delta_j) \right]^{\sigma_j/(1-\sigma_j)} + W_K \frac{1}{\phi_j} Q_j \left[\delta_j \left(\frac{(1-\delta_j)}{\delta_j} \right)^{(1-\sigma_j)} + (1-\delta_j) \right]^{\sigma_j/(1-\sigma_j)}$$

With the assumption that the benchmark equilibrium market prices (P_j), and factor prices (W_L , and W_K) are equal to 1, then we can solve for ϕ_j

$$\phi_j = \left((\delta_j)^{\sigma_j} + (1-\delta_j)^{\sigma_j} \right)^{1/(1-\sigma_j)}.$$

The determination of the parameter values in the consumer side follows a similar procedure. Recall, the consumer utility function,

$$U_h = \sum_j \left[(\alpha_{jh})^{1/\sigma_h} (X_{jh})^{(\sigma_h-1)/\sigma_h} \right]^{\sigma_h/(\sigma_h-1)},$$

but only the expenditure share for consumption spending on goods j for household h , α_{jh} , is calculated,

$$\alpha_{jh} = \left(\frac{(\omega_h^j)}{((\theta_h^{jj'})^{1-\sigma_h} + \omega_h^j)} \right)$$

where ω_h^j is the ratio of the household share of total expenditure on goods j and j' ,

$$\omega_h^j = P_j X_{jh} / P_{j'} X_{j'h}$$

and $\theta_h^{jj'}$ is the price ratio of good j and j' , $\theta_h^{jj'} = P_j / P_{j'}$. The elasticity of substitution in the

household, σ_h , is generally obtained from the literature which uses the own-price elasticity as an approximation as shown in Shoven and Whalley (1992).

These parameter values once generated can be used in further model applications to solve for alternative equilibria which result from policy changes.

Full GAMS Model

```

SETS      Factor          Basic factors of production    /Labor, Capital/
          Sector          Producing industries          /Food, NonFood/
          households      Households Types            /Farm, NonFarm/ ;
ALIAS (households,households1), (Sector,OtherSector), (Factor,Factor1);

PARAMETER SigmaC(households)  Elasticity of substitution in household CES
/NonFarm 1.5
Farm 1.00/ ;
TABLE Alpha(Sector,households)  Consumption share in household CES
      NonFarm  Farm
Food  0.5     0.4
NonFood 0.5   0.6 ;
TABLE Endowment(Factor,households)  Household endowments of factors
      NonFarm  Farm
Labor  60     5
Capital 25    2 ;
Table Intermediateuse(OtherSector,sector) Use in othersector when producing in sector
      Nonfood  Food
Nonfood 0.00  0.0
Food 0.00  0.0;
PARAMETER Phi(Sector)  scale parameter in CES production function
/NonFood 2.0/ ;
TABLE Delta(Factor,Sector)  distribution parameter in CES production
      Food  NonFood
Labor  0.6  0.7
Capital 0.4 0.3 ;
PARAMETER Sigma(Sector)  Elasticity of production factor substitution
/NonFood 0.5/
Parameter TaxShare(households)  Household share of tax disbursements
/NonFarm 0.7
Farm 0.3/ ;
Parameter Governmentpurch(sector)  Government goods purchase dependence on revenue
/NonFood 0.00
NonFood 0.00 /
Scalar Incometax Household tax level /0.001/;
Parameter CorporateTax(Sector)  Tax on corporate earnings
/NonFood 0, nonfood 0/;
Parameter TaxDeduction(Households) Household Tax Deductions
/farm 0, NonFarm 0/;

POSITIVE VARIABLES
FACTORPRICE(Factor)          Prices for the factors labor and capital
FACTORQUAN(Factor,Sector)    Factor use by a producing sector
COMPRICE(Sector)             Prices of commodities
DemCommod(Households,Sector) Commodity Demand by Households
PRODUCTION(Sector)           Production Quantity
HHIncome(households)         Household income
TAXREVENUE                    Total government tax revenue ;

EQUATIONS
FactorMkt(Factor)            Factor market balances
FactorDem(Factor,Sector)     Factor demand by a sector
CommodMkt(Sector)            Commodity market balance
CommodDem(Households,Sector) Commodity Demand by Households
Profit(Sector)                Zero profit condition
Income(households)           Household budget constraint
GovBal                        Government budget constraint ;

FactorMkt(Factor)..
sum(households,Endowment(Factor,households)) =G=
sum(Sector,FactorQuan(Factor,Sector)) ;

FactorDem(Factor,Sector)..
FACTORQUAN(Factor,Sector) =g=
Production(Sector)*Phi(Sector)**(sigma(Sector)-1)
*(Delta(Factor,Sector))

```

```

* ( sum( Factor1, Delta( Factor1, Sector) ** sigma( Sector)
      * ( FactorPrice( Factor1) ) ** ( 1 - sigma( Sector) ) )
  ** ( 1 / ( 1 - sigma( Sector) ) ) / Phi( Sector) ) /
  FactorPrice( Factor ) ** sigma( Sector) ;

CommodDem( Households, Sector ) ..
  DemCommod( Households, Sector ) =E=
  ( HHIncome( households )
    / sum( otherSector, alpha( otherSector, households )
      * ComPrice( OtherSector ) ** ( 1 - SigmaC( households ) ) )
    ) * Alpha( Sector, households ) * ( 1 / ComPrice( Sector ) ) ** sigmaC( households );

CommodMkt( Sector ) ..
  Production( Sector ) =G=
  sum( households, DemCommod( Households, Sector ) )
+ Governmentpurch( sector ) * TaxRevenue / COMPRICE( Sector )
+ sum( othersector, intermediateuse( sector, othersector ) * production( othersector ) );

Profit( Sector ) ..
  sum( Factor, FactorPrice( Factor ) * FactorQuan( Factor, Sector ) )
+ corporatetax( sector ) * (
  ComPrice( Sector ) * Production( Sector )
  - sum( othersector, ComPrice( OtherSector ) * intermediateuse( othersector, sector )
    * production( sector ) )
  - sum( Factor, FactorPrice( Factor ) * FactorQuan( Factor, Sector ) ) )
=G= ComPrice( Sector ) * Production( Sector ) ;

Income( households ) ..
  HHIncome( households ) =G=
  ( 1 - incometax ) * sum( Factor, Endowment( Factor, households ) * FactorPrice( Factor ) )
+ incometax * TaxDeduction( Households )
+ TaxShare( households ) * TaxRevenue ;

GovBal ..
  TaxRevenue =G=
  sum( sector, corporatetax( sector ) * (
    ComPrice( Sector ) * Production( Sector )
    - sum( Othersector, ComPrice( OtherSector ) * intermediateuse( Othersector, sector )
      * production( sector ) )
    - sum( Factor, FactorPrice( Factor ) * FactorQuan( Factor, Sector ) ) ) )
+ sum( households, incometax
  * ( sum( Factor, Endowment( Factor, households ) * FactorPrice( Factor ) )
    - TaxDeduction( Households ) ) );

MODEL CGEModel /FactorMkt.FACTORPRICE, FactorDem.FACTORQUAN,
CommodMkt.COMPRICE, commoddem.DemCommod
Profit.PRODUCTION, Income.Hhincome, GovBal.TAXREVENUE/;

* lower bounds
  FACTORPRICE.LO( Factor ) = 0.01;
  COMPRICE.LO( Sector ) = 0.01;
  HHincome.Lo( households ) = 0.01;
*starting point
  FACTORPRICE.L( Factor ) = 1 ;
  FACTORQUAN.L( Factor, sector ) = 1 ;
  DemCommod.l( Households, Sector ) = 1;
  COMPRICE.L( Sector ) = 1 ;
  PRODUCTION.L( Sector ) = 10;
  HHincome.L( households ) =
  sum( Factor, FACTORPRICE.l( Factor ) * Endowment( Factor, households ) );
  HHincome.fx( households ) $( ord( households ) = 1 ) = HHincome.L( households );

OPTION MCP = PATH;
SOLVE CGEModel USING MCP;

```

Reference

- Creedy, J. *General Equilibrium and Welfare*. Edward Elgar Publishing Limited, Cheltenham, UK, 1996.
- Ferris, M. C. and T. S. Munson. "GAMS/PATH User Guide Version 4.3."
<http://www.gams.com/solvers/solvers.htm#PATH>, 2000.
- Ferris, M. C. and J. S. Pang. "Engineering and Economic Applications of Complementarity Problems." *SIAM Review*, 39:669-713, 1997.
- Manne, A. S. *Economic Equilibrium: Model Formulation and Solution*. Mathematical Programming Study 23, edited by A. S. Manne, A publication of the Mathematical Programming Society, Amsterdam, the Netherlands, 1985.
- Robinson, S. and M. El-Said, *GAMS Code for Estimating A Social Accounting Matrix (SAM) Using Cross Entropy (CE) Methods*, Trade and Macroeconomics Division Discussion Papers, Number 64, Dec 2000, <http://www.ifpri.org/divs/tmd/dp/tmdp64.htm>
- Shoven, J. B. and J. Whalley. "General-Equilibrium with Taxes: A Computational Procedure and an Existence Proof." *The Review of Economic Studies*, 40:475-489, 1973.
- Shoven, J. B. and J. Whalley. "Applied General-Equilibrium Models of Taxation and International Trade: An Introduction and Survey." *J. Economic Literature*, 22:1007-1051, 1984.
- Shoven, J. B. and J. Whalley. *Applying General Equilibrium*. Cambridge University Press, New York, 1998.
- Varian, H. R. *Microeconomic Analysis*. W. W. Norton & Company, Inc., New York, 1992